253 CS 701 Fall 2003^©

Reading Assignment

• Read Section 15.5, “Automatic

Instruction Selection,” from Chapter 15.

• ReadPelegri-LlopartandGraham’spaper, “Optimal Code Generation from Expression Trees.” (Linked from the class Web page.)

• Read Fraser, Henry and Proebsting’s paper, “BURG--Fast Optimal Instruction Selection and Tree Parsing.” (Linked from the class Web page.)

254 CS 701 Fall 2003^©

Software Pipelining

Often loop bodies are too small to allow effective code scheduling. But loop bodies, being “hot spots,” are exactly where scheduling is most important. Consider

void f (int a[],int last) {
  for (p=&a[0];p!=&a[last];p++)
     (*p)++;
}

The body of the loop might be:

L: ld   [%g3],%g2
   nop
   add  %g2,1,%g2
   st   %g2,[%g3]
   add  %g3,4,%g3
   cmp  %g3,%g4
   bne  L
   nop

255 CS 701 Fall 2003^©

Scheduling this loop body in isolation is ineffective—each instruction depends upon its immediate predecessor. So we have a loop body that takes 8 cycles to execute 6 “core” instructions.

We could unroll the loop body, but for how many iterations? What if the loop ends in the “middle” of an expanded loop body? Will extra registers be a problem?

256 CS 701 Fall 2003^©

In this casesoftware pipelining offers a nice solution. We expand the loop body symbolically, intermixing instructions from several iterations. Instructions can overlap, increasing parallelism and forming a “tighter” loop body:

   ld   [%g3],%g2
   nop
   add  %g2,1,%g2
L: st   %g2,[%g3]
   add  %g3,4,%g3
   ld   [%g3],%g2
   cmp  %g3,%g4
   bne  L
   add  %g2,1,%g2

Now the loop body is ideal—exactly 6 instructions. Also, no extra registers are needed! But, we do “overshoot” the end of the loop a bit, loading one element past the exit point. (How serious is this?)

257 CS 701 Fall 2003^©

Key Insight of Software Pipelining

Software pipelining exploits the fact that a loop of the form{A B C}^n, where A, B andC are individual instructions, andnistheiterationcount,isequivalent to A{BCA}^{n-1}BCandisalsoequivalent to A B {C A B}^{n-1} C. Mixing instructions from several iterationsmayincreasethe effectiveness of code scheduling, and may perhaps allow for more parallel execution.

Software Pipelining is Hard

In fact, it is NP-complete:

HsuandDavidson,“Highlyconcurrent scalar processing,” 13th ISCA (1986). 258 CS 701 Fall 2003^©

The Iteration Interval

We seek to initiate the next iteration of a loop as soon as possible, squeezing each iteration of the loop body into as few machine cycles as possible. The general form of a software pipelined loop is:

Prologue Code

Kernel Code

Epilogue Code

259 CS 701 Fall 2003^©

Theprologuecode“setsup”the main loop, and the epilogue code “cleans up” after loop termination. Neither the prolog nor the epilogue need be optimized, since they execute only once. Optimizing the kernel is key in software pipelining. The kernel’s execution time (in cycles) is called the initiation interval (II); it measures how quickly the next iteration of a loop can start. We want the smallest possible initiation interval. Determining the smallest viable II is itself NPcomplete. Because of parallel issue and execution in superscalar and multipleissueprocessors,verysmallII values are possible (even less than 1!)

260 CS 701 Fall 2003^©

Factors that Limit the Size of the Initiation Interval

We want the initiation interval to be assmallaspossible.Twofactorslimit how small the II can become:

• Resource Constraints

• Dependency Constraints

261 CS 701 Fall 2003^©

Resource Constraints

AsmallIInormallymeansthat weare doing steps of several iterations simultaneously. The number of registers and functional units (that execute instructions) can become limiting factors of the size of II. Forexample,ifa loopbodycontains4 floating point operations, and our processor can issue and execute no morethan2 floatingpointoperations per cycle, then the loop’s II can’t be less than 2.

262 CS 701 Fall 2003^©

Dependency Constraints

A loop body can often contain aloopcarried dependence. This means one iteration of a loop depends on values computed in an earlier iteration. For example, in

void f (int a[]) {
  for (i=1;i<1000;i++)
     a[i]=(a[i-1]+a[i])/2;
}

there is a loop carried dependence from theuseofa[i-1] to thecomputationof

a[i] in the previous iteration. This

means the computation ofa[i] can’t begin until the computation ofa[i-1] is completed. Let’s look at the code that might be generated for this loop:

263 CS 701 Fall 2003^©

f:
    mov %o0, %o2 !a in %o2
    mov 1, %o1         !i=1 in %o1
L:
    sll   %o1, 2, %o0    !i*4 in %o0
    add   %o0, %o2, %g2  !&a[i] in %g2
ld [%g2-4], %g2   !a[i-1] in %g2
    ld [%o2+%o0], %g3 !a[i] in %g3
add %g2, %g3, %g2 !a[i-1]+a[i]
srl %g2, 31, %g3 !s=0 or 1=sign
add   %g2, %g3, %g2 !a[i-1]+a[i]+s
sra %g2, 1, %g2 !a[i-1]+a[i]/2
    add %o1, 1, %o1 !i++
    cmp %o1, 999
    ble L
st %g2, [%o2+%o0] !store a[i]
    retl
    nop

The 6 marked instructions form a cyclic dependency chain from a use ofa[i-1] to its computation (asa[i]) in the previouscycle.Thiscyclemeansthat the loop’s II can never be less than 6. 264 CS 701 Fall 2003^©

Modulo Scheduling

There are many approaches to software pipelining. One of the simplest, and best known, ismodulo scheduling. Modulo scheduling builds upon the postpass basic block schedulers we’ve already studied. First, we estimate the II of the loop we will create. How? We can compute the minimum II based on resource considerations (II_{res}) and the minimum II based on cyclic loop-carried dependencies (II_{dep}). Then max(II_{res},II_{dep}) is a reasonable estimate of the best possible II. We’ll try to build a loop with a kernel size of II. If this fails, we’ll try II+1, II+2, etc.

265 CS 701 Fall 2003^©

In modulo scheduling we’ll schedule instructions one by one, using the dependency dag and whatever heuristic we prefer to choose among multiple roots. Now though, if we place an instruction at cycle c (many independentinstructionsmayexecute in the same cycle), then we’ll place additionalcopiesoftheinstructionat cycle c+II, c+2*II, etc. Placement must respect dependency constraints and resource limits at all positions. We consider placements only until a kernel (of size II) forms. The kernel must begin before cycle s- 1,wheresisthe sizeofthe loopbody (in instructions). The loop’s conditionalbranchisplacedafterthe kernel is formed. 266 CS 701 Fall 2003^©

If we can’t form a kernel of size II (because of dependency or resource conflicts), we increase II by 1 and try again.Atworst,wegeta kernelequal in size to the original loop body, which guarantees that the modulo scheduler eventually terminates. Depending on how many iterations areintermixedinthe kernel,the loop terminationconditionmayneedto be adjusted (since the initial and final iterations may appear as part of the loop prologue and epilogue).

267 CS 701 Fall 2003^©

Example

Consider the following simple functionwhichaddsanarrayindexto each element of an array and copies the results into a second array:

void f (int a[],int b[]) {
  t1 = &a[0];
  t2 = &b[0];
  for (i=0;i<1000;i++,t1++,t2++)
     *t1 = *t2 + i;
}

The code forf (compiled as a leaf procedure) is:

268 CS 701 Fall 2003^©

1.   f:   mov     0, %g3
2.   L:   ld      [%o1], %g2
3.        add     %g3, %g2, %g4
4.        st      %g4, [%o0]
5.        add     %g3, 1, %g3
6.        add     %o0, 4, %o0
7.        cmp     %g3, 999
8.        ble     L
9.        add     %o1, 4, %o1
10.       retl
11.       nop

                    2

              3 9

        4 5

        6 7

Dashed arcs are anti dependencies.

269 CS 701 Fall 2003^©

We’llsoftwarepipelinethe loopbody, excluding the conditional branch (whichisplacedafterthe loopkernel is formed). This loop body contains 2 loads/ stores, 5 arithmetic and logical operations (including the compare) and one conditional branch. Let’s assume the processor we are compilingforhas1 load/storeunit,3 arithmetic/logic units, and 1 branch unit. That means the processor can (ideally) issue and execute simultaneously 1 load or store, 3 arithmeticandlogicinstructions, and 1 branch. Thus its maximum issue widthis5. (Currentsuperscalarshave roughly this capability.)

270 CS 701 Fall 2003^©

Considering resource requirements, we will need at least two cycles to processthecontentsoftheloopbody. There are no loop-carried dependencies. Thuswewillestimatethisloop’s best possible Initiation Interval to be 2. Since the only instruction that can stall is the root of the dependency dag, we’ll schedule using estimated critical path length, which is just the node’s heightin the tree.Hencewe’ll schedule the nodes in the order: 2,3,4,5,6,7,9. We’ll schedule all instructions in a legal execution order (respecting dependencies),andwe’lltryto choose as many instructions as possible to execute in the same cycle.

271 CS 701 Fall 2003^©

Starting with the root, instruction 2, we schedule it at cycles 1, 3 (=1+II), 5 (=1+2*II):

cycle     instruction

1.           ld    [%o1], %g2
2.
3.           ld    [%o1], %g2
4.
5.           ld    [%o1], %g2

No conflicts so far, since each of the loads starts an independent iteration.

272 CS 701 Fall 2003^©

We’ll schedule instruction 3 next. It must be placed at cycles 3, 5 and 7 since it uses the result of the load.

cycle     instruction

1.           ld    [%o1], %g2
2.
3.           add   %g3, %g2, %g4
3.           ld    [%o1], %g2
4.
5.           add   %g3, %g2, %g4
5.           ld    [%o1], %g2
6.
7.           add   %g3, %g2, %g4

Note that in cycles 3 and 5 we use the current value of%g2 and initiate a load into%g2.

273 CS 701 Fall 2003^©

Instruction4 isnext.Itusestheresult of the add we just scheduled, so it is placed at cycles 4 and 6.

   cycle     instruction

   1.           ld    [%o1], %g2
   2.
   3.           add   %g3, %g2, %g4
   3.           ld    [%o1], %g2
   4.           st    %g4, [%o0]
   5.           add   %g3, %g2, %g4
   5.           ld    [%o1], %g2
   6.           st    %g4, [%o0]
   7.           add   %g3, %g2, %g4

274 CS 701 Fall 2003^©

Instruction 5 is next. It is anti dependentoninstruction3,sowecan placeit inthe samecyclesthat 3 uses (3, 5 and 7).

  cycle     instruction

   1.           ld    [%o1], %g2
   2.
   3.           add   %g3, %g2, %g4
   3.           ld    [%o1], %g2
   3.           add   %g3, 1, %g3
   4.           st    %g4, [%o0]
   5.           add   %g3, %g2, %g4
   5.           ld    [%o1], %g2
   5.           add   %g3, 1, %g3
   6.           st    %g4, [%o0]
   7.           add   %g3, %g2, %g4
   7.           add   %g3, 1, %g3

275 CS 701 Fall 2003^©

Instruction 6 is next. It is anti dependentoninstruction4,sowecan placeit inthe samecyclesthat 4 uses (4 and 6).

   cycle     instruction

   1.           ld    [%o1], %g2
   2.
   3.           add   %g3, %g2, %g4
   3.           ld    [%o1], %g2
   3.           add   %g3, 1, %g3
   4.           st    %g4, [%o0]
   4.           add   %o0, 4, %o0
   5.           add   %g3, %g2, %g4
   5.           ld    [%o1], %g2
   5.           add   %g3, 1, %g3
   6.           st    %g4, [%o0]
   6.           add   %o0, 4, %o0
   7.           add   %g3, %g2, %g4
   7.           add   %g3, 1, %g3

276 CS 701 Fall 2003^©

Next we place instruction 7. It uses the resultofinstruction5 (%g3),soit is placed in the cycles following instruction 5 (4 and 6).

  cycle     instruction

   1.           ld    [%o1], %g2
   2.
   3.           add   %g3, %g2, %g4
   3.           ld    [%o1], %g2
   3.           add   %g3, 1, %g3
   4.           st    %g4, [%o0]
   4.           add   %o0, 4, %o0
   4.           cmp   %g3, 999
   5.           add   %g3, %g2, %g4
   5.           ld    [%o1], %g2
   5.           add   %g3, 1, %g3
   6.           st    %g4, [%o0]
   6.           add   %o0, 4, %o0
   7.           add   %g3, %g2, %g4
   7.           add   %g3, 1, %g3
   7.           cmp   %g3, 999

277 CS 701 Fall 2003^©

Finally we place instruction 9. It is anti dependent on instruction 2 so it is placed in the same cycles as instruction 3 (1, 3 and 5).

   cycle       instruction
   1.           ld    [%o1], %g2
   1.           add   %o1, 4, %o1
   3.           add   %g3, %g2, %g4
   3.           ld    [%o1], %g2
   3.           add   %o1, 4, %o1
   3.           add   %g3, 1, %g3
   4.           st    %g4, [%o0]
   4.           add   %o0, 4, %o0
   4.           cmp   %g3, 999
   5.           add   %g3, %g2, %g4
   5.           ld    [%o1], %g2
   5.           add   %o1, 4, %o1
   5.           add   %g3, 1, %g3
   6.           st    %g4, [%o0]
   6.           add   %o0, 4, %o0
   7.           add   %g3, %g2, %g4
   7.           add   %g3, 1, %g3
   7.           cmp   %g3, 999

278 CS 701 Fall 2003^©

We look for a 2 cycles kernel that containsall7 instructionsofthe loop bodythat wehavescheduled.Wealso want a kernel that sets the condition code (via the cmp) during its first cycle so that it can be tested during its second (and final) cycle. Cycles 4 and 5 meet these criteria, and will form our kernel. We place the conditional branch just before the last instruction in cycle 5 (to give the conditional branch a useful instruction for its delay slot).

279 CS 701 Fall 2003^©

We now have:

   cycle       instruction
   1.           ld    [%o1], %g2
   1.           add   %o1, 4, %o1
   3.           add   %g3, %g2, %g4
   3.           ld    [%o1], %g2
   3.           add   %o1, 4, %o1
   3.           add   %g3, 1, %g3
   4.      L:   st    %g4, [%o0]
   4.           add   %o0, 4, %o0
   4.           cmp   %g3, 999
   5.           add   %g3, %g2, %g4
   5.           ld    [%o1], %g2
   5.           add   %o1, 4, %o1
   5.           ble   L
   5.           add   %g3, 1, %g3
   6.           st    %g4, [%o0]
   6.           add   %o0, 4, %o0
   7.           add   %g3, %g2, %g4
   7.           add   %g3, 1, %g3
   7.           cmp   %g3, 999

280 CS 701 Fall 2003^©

Acoupleoffinalissuesmustbedealt with:

• Does the iteration count need to be changed? In this case no, since the final valid value ofi, 999, is used to compute

 %g4 in cycle 5, before the loop exits.
• What instructions do we keep as the

loop’s epilogue? None! Instructions past the kernel aren’t needed since they are part of futureiterations(pasti==999)which aren’t needed or wanted.

• Notethat b[1000] andb[1001] are

“touched”eventhoughtheyarenever used. This is probably OK as long as arraysaren’tplacedat theveryendof a page or segment.

281 CS 701 Fall 2003^©

Our final loop is:

   cycle       instruction
   1.           ld    [%o1], %g2
   1.           add   %o1, 4, %o1
   3.           add   %g3, %g2, %g4
   3.           ld    [%o1], %g2
   3.           add   %o1, 4, %o1
   3.           add   %g3, 1, %g3
   4.      L:   st    %g4, [%o0]
   4.           add   %o0, 4, %o0
   4.           cmp   %g3, 999
   5.           add   %g3, %g2, %g4
   5.           ld    [%o1], %g2
   5.           add   %o1, 4, %o1
   5.           ble   L
   5.           add   %g3, 1, %g3

Thisisveryefficientcode—weusethe full parallelism of the processor, executing 5 instructions in cycle 5 and8 instructionsinjust2 cycles.All resource limitations are respected.